Calculating organism heading

Determining the “front” directions from the positions of the cells’s neighbors - Joakim Johansson

Two dimensions

First we need a way to plot vectors:

In[1]:=
```
PlotVector[l_] := Graphics[Arrow[{{0, 0}, #}] & /@ l, PlotRange → {{-2, 2}, {-2, 2}},
   Axes → True];
```

Calculate base vectors of transformed base from neighbor vectors. (And thereby transformation matrix)

In[2]:=
```
getTransform[right_, front_, left_, back_] :=
 {
  Normalize[right - left],
  Normalize[front - back]
 }
```

Initialize neighbor vectors as rotated $\frac{\pi}{3}$ radians:

In[5]:=
```
{rotRight, rotFront, rotLeft, rotBack} = RotationMatrix[π/3].# & /@ {
    {1, 0}, {0, 1}, {-1, 0}, {0, -1}
   };
PlotVector[{rotRight, rotFront, rotLeft, rotBack}]
PlotVector[{#}] & /@ {rotRight, rotFront, rotLeft, rotBack}
```

Out[6]=



Out[7]=



Calculate transformational matrix M:

In[8]:=
```
M = getTransform[rotRight, rotFront, rotLeft, rotBack];
M // MatrixForm
PlotVector[M // Transpose]
```

Out[9]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2} & \frac{\sqrt{3}}{2} \\ -\frac{\sqrt{3}}{2} & \frac{1}{2} \end{pmatrix}$$

Out[10]=



In[11]:=
```
Manipulate[PlotVector[{{x, y}, M.{x, y}}], {x, -1, 1}, {y, -1, 1}]
```

Out[11]=



Three dimensions

First we need a way to plot vectors in 3d:

In[3]:=
```
PlotVector3D[l_] := Graphics3D[Arrow[{{0, 0, 0}, #}] & /@ l,
   Axes → True];
```

Calculate base vectors of transformed base from neighbor vectors. (And thereby transformation matrix)

In[4]:=
```
getTransform3D[xPlus_, xMinus_, yPlus_, yMinus_, zPlus_, zMinus_] :=
 {
  Normalize[xPlus - xMinus],
  Normalize[yPlus - yMinus],
  Normalize[zPlus - zMinus]
 }
```

Initialize neighbor vectors as slightly perturbed:

In[12]:=
```
{xPlus, xMinus, yPlus, yMinus, zPlus, zMinus} = (0.2 RandomReal[] + #) & /@
  {
   {1, 0, 0}, {-1, 0, 0},
   {0, 1, 0}, {0, -1, 0},
   {0, 0, 1}, {0, 0, -1}
  };
PlotVector3D[{xPlus, xMinus, yPlus, yMinus, zPlus, zMinus}]
```

Out[13]=



Calculate transformational matrix M:

In[14]:=
```
M = getTransform3D[xPlus, xMinus, yPlus, yMinus, zPlus, zMinus];
M // MatrixForm
PlotVector3D[M // Transpose]
```

Out[15]//MatrixForm=

$$\begin{pmatrix} 0.99884 & 0.0340479 & 0.0340479 \\ -0.0470861 & 0.99778 & -0.0470861 \\ -0.0426775 & -0.0426775 & 0.998177 \end{pmatrix}$$

Out[16]=



In[17]:=
```
Manipulate[PlotVector3D[{{x, y, z}, M.{x, y, z}}], {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```

Out[17]=



Finally, what is the rule for dot product now again?

In[18]:=
```
{
  {a.x, a.y, a.z},
  {b.x, b.y, b.z},
  {c.x, c.y, c.z}
 }.{v.x, v.y, v.z}
```

Out[18]= {a.x v.x + a.y v.y + a.z v.z, b.x v.x + b.y v.y + b.z v.z, c.x v.x + c.y v.y + c.z v.z}